

## Tech Note 815

### Using the ShowGraphic() Function with ArcestrA Graphics

---

All Tech Notes, Tech Alerts and KBCD documents and software are provided "as is" without warranty of any kind. See the [Terms of Use](#) for more information.

Topic#: 002605  
Created: January 2012

#### Introduction

This tech note provides examples of how to use the **ShowGraphic()** function in ArcestrA Graphics. This function is available within any ArcestrA Graphic client script.

#### Application Versions

- Wonderware InTouch® 10.5 and later
- Wonderware Application Server 3.5 and later

#### Examples

This *Tech Note* includes the following examples. Each section contains script samples you can copy/paste into your Objects.

- [Show/Hide Graphic](#)
- [Show Graphic in a Specific Location](#)
- [Show Multiple Graphics with One Button Click](#)
- [Show Graphic Using a Custom Property Value](#)

#### Examples

##### Show/Hide Graphic

This example shows how to use the **ShowGraphic()** with some of its properties.

1. Create an ArcestrA Graphic Symbol on the Graphic toolbox (e.g. Symbol\_001).
2. Embed a button and name it **ShowGraphic**.
3. Select an Animation Action Script and type the script **On Left Click/Key Down** as shown below:

```
Dim graphicInfo as aaGraphic.GraphicInfo;
graphicInfo.Identity = "MyMooCow";
graphicInfo.GraphicName = "MooCow";
graphicInfo.Resizable = true;
graphicInfo.WindowRelativePosition = 1; '(ENUM value of 1 means using Window area)
graphicInfo.WindowLocation = 8; '(ENUM value of 8 is Rightof the above window relative position)
graphicInfo.ScalePercentage = 150;
ShowGraphic( graphicInfo );
```





FIGURE 1: ACTION SCRIPT ON LEFT CLICK/KEY DOWN

5. Embed another button and name it **HideGraphic**.
6. Select an Animation Action Script and type the script for type **On Left Click/Key Down**.



**FIGURE 2: HIDEGRAPHIC( "MyMooCow");**

7. Create a new derived InTouchViewApp.
8. Create an InTouch Window call it **Win1**.
9. Embed the Symbol on the InTouch Window (Figure 3 below).




FIGURE 3: SHOW/HIDE SYMBOL ON TEST WINDOW

10. Switch to Runtime mode.
11. Click the **ShowGraphic** button. The **MyMooCow** graphic displays on the right side of the window as specified in the script (Figure 4 below).




FIGURE 4: SHOW/HIDE GRAPHIC IN RUNTIME

- Click the **HideGraphic** button and the graphic disappears.

### Show Graphic in a Specific Location

This example shows how to use **ShowGraphic()** so you can place the symbol in a specific location.

- Create an Arcestra Graphic Symbol on the Graphic toolbox (e.g. **Symbol\_002**)
- Embed a button and name it **ShowGraphic**.
- Embed another button and name it **HideGraphic**.
- Embed a rectangle called **Boundary** on the right side (Figure 5 below).




FIGURE 5: BLACK RECTANGLE "BOUNDARY"

- Select an Animation Action Script on the ShowGraphic button and type the script **On Left Click/Key Down**.

```
Dim graphicInfo as aaGraphic.GraphicInfo:
graphicInfo.Identity = "MyTank";
graphicInfo.GraphicName = "SSTankVertPill";
graphicInfo.WindowLocation = aaGraphic.WindowLocation.TopLeftCorner;
graphicInfo.WindowRelativePosition = aaGraphic.WindowRelativePosition.ClientAreaXY;
```

```
graphicInfo.RelativeTo = aaGraphic.RelativeTo.CustomizedWidthHeight;  
graphicInfo.X = Boundary.X;  
graphicInfo.Y = Boundary.Y;  
graphicInfo.Height = Boundary.Height;  
ShowGraphic( graphicInfo );
```




FIGURE 6: SHOW/HIDE WITH LOCATION

6. Select an Animation Action Script and type the script as shown below **On Left Click/Key Down**.




FIGURE 7: `HIDEGRAPHIC("MYTANK");`

7. Create another InTouch Window and call it **Win2**.
8. Embed the Symbol on the InTouch Window (Figure 8 below).




FIGURE 8: WIN2 GRAPHIC

9. Switch to Runtime.
10. Click on the **ShowGraphic** button. The **MyTank** graphic is displayed on the black rectangle where you specified the location (Figure 9 below).






FIGURE 9: Win2 RUNTIME WITH LOCATION

11. Click on the **HideGraphic** button to hide the **MyTank** graphic.

### Show Multiple Graphics with One Button Click

This example shows how to use the ShowGraphic() function to show multiple symbols with just one button click.

1. Create an ArcestrA Graphic Symbol on the Graphic toolbox (**Symbol\_003**).
2. Embed a button and name it **Multiple ShowGraphics**.
3. Embed another button and name it **HideAll Graphics**.
4. Embed a rectangle named Boundary (Figure 10 below).




FIGURE 10: SYMBOL\_003 "MULTIPLE GRAPHICS"

5. Select an **Animation Action Script** on the Multiple ShowGraphics button and copy/paste the following script for **On Left Click/Key Down** (Figure 11 below)

```
Dim i;
For i = 1 TO 8

Dim graphicInfo as aaGraphic.GraphicInfo;
graphicInfo.Identity = "ShowAll" + StringFromIntg( i, 1);
graphicInfo.WindowLocation = aaGraphic.WindowLocation.TopLeftCorner;
graphicInfo.WindowRelativePosition = aaGraphic.WindowRelativePosition.ClientAreaXY;
graphicInfo.RelativeTo = aaGraphic.RelativeTo.CustomizedWidthHeight;

IF i == 1 THEN
graphicInfo.GraphicName = "AnalogMeter90DegreeBlack";
graphicInfo.WindowTitle = "Clock";
graphicInfo.X = Boundary.X;
graphicInfo.Y = Boundary.Y;
ENDIF;

IF i == 2 THEN
graphicInfo.GraphicName = "ClockWonderware";
graphicInfo.WindowTitle = "Clock";
graphicInfo.X = Boundary.X + (Boundary.Width/4);
graphicInfo.Y = Boundary.Y;
ENDIF;

IF i == 3 THEN
graphicInfo.GraphicName = "PumpCentrifugal";
graphicInfo.WindowTitle = "Pump";
graphicInfo.X = Boundary.X + (2 * (Boundary.Width/4));
graphicInfo.Y = Boundary.Y;
ENDIF;

IF i == 4 THEN
graphicInfo.GraphicName = "ValveAirBall";
graphicInfo.WindowTitle = "Valve";
graphicInfo.X = Boundary.X + (3 * (Boundary.Width/4));
graphicInfo.Y = Boundary.Y;
ENDIF;

IF i == 5 THEN
graphicInfo.GraphicName = "WonderButton";
graphicInfo.WindowTitle = "Button";
graphicInfo.X = Boundary.X;
graphicInfo.Y = Boundary.Y + (Boundary.Height / 2);
ENDIF;

IF i == 6 THEN
graphicInfo.GraphicName = "Flames";
graphicInfo.WindowTitle = "Flame";
graphicInfo.X = Boundary.X + (Boundary.Width/4);
graphicInfo.Y = Boundary.Y + (Boundary.Height/2);
ENDIF;

IF i == 7 THEN
graphicInfo.GraphicName = "SkullCrossbonesSign";
graphicInfo.WindowTitle = "Skull";
graphicInfo.X = Boundary.X + (2 * (Boundary.Width/4));
graphicInfo.Y = Boundary.Y + (Boundary.Height/2);
ENDIF;

ENDIF;
```

```

IF i == 8 THEN
    graphicInfo.GraphicName = "HappyFace";
    graphicInfo.WindowTitle = "Smiley Face";
    graphicInfo.X = Boundary.X + (3 * (Boundary.Width/4));
    graphicInfo.Y = Boundary.Y + (Boundary.Height/2);
ENDIF;

graphicInfo.Width = (Boundary.Width/4);
graphicInfo.Height = (Boundary.Height/2);
ShowGraphic( graphicInfo );
NEXT;

```




FIGURE 11: ACTION SCRIPT FOR ON LEFT CLICK/KEY DOWN

6. Select an Animation Action Script on the **HideAll Graphics** button, and copy/paste the following script for **On Left Click/Key Down** (Figure 12 below).

```

Dim i;
For i = 1 to 8
HideGraphic( "ShowAll" + StringFromIntg( i, 1 ) );
Next;

```




FIGURE 12: ACTION SCRIPT FOR HIDEGRAPHIC ( )

7. Create another InTouch Window call it **Win3**.
8. Embed the above Symbol on the InTouch Window (Figure 13 below).




FIGURE 13: WIN3 INTOUCH WINDOW WITH MULTIPLE GRAPHICS BUTTONS/PANEL

9. Switch to Runtime.
10. Click the **Multiple ShowGraphics** button. All 8 Arcestra Graphics are displayed on the black rectangle where you specified the location (Figure 14 below).




FIGURE 14: MULTIPLE SHOWGRAPHICS: RUNTIME

11. Click the **HideAll Graphics** button and all the 8 ArcestrA Graphics will disappear.

### Show Graphic Using a Custom Property Value

This example shows how to use the **ShowGraphic()** function in which you can change the custom property value.

1. Create an ArcestrA Graphic Symbol on the Graphic toolbox (**Symbol\_004**).
2. Embed a button and name it **ShowGraphic**.
3. Embed another button and name it **HideGraphics**.
4. Embed a rectangle named **Boundary** on the right side (Figure 15 below).




FIGURE 15: InTOUCH w/SHOW - HIDE GRAPHIC + RECTANGLE AREA

5. Create an InTouch tag called **ValueTag** of type **Memory Integer**.
6. Create an application script while running every 1000ms as shown below so that we can change the value of the meter every second (Figure 16 below).

```

IF ValueTag > 110
THEN
  ValueTag = 10;
ELSE
  ValueTag = ValueTag + 1;
ENDIF;

```




FIGURE 16: WHILE RUNNING APPLICATION SCRIPT



7. Select an Animation Action Script on the **ShowGraphic** button and copy/paste the script for **On Left Click/Key Down** (Figure 17 below).

```
Dim graphicInfo as aaGraphic.GraphicInfo;
Dim cpValues [3] as aaGraphic.CustomPropertyValuePair;
graphicInfo.Identity = "MyAnalogMeter";
graphicInfo.GraphicName = "AnalogMeter180Degree";
cpValues[1] = new aaGraphic.CustomPropertyValuePair("Max", 110, true);
cpValues[2] = new aaGraphic.CustomPropertyValuePair("Min", 10, true);
cpValues[3] = new aaGraphic.CustomPropertyValuePair("Value", "InTouch:ValueTag", false);
graphicInfo.CustomProperties = cpValues;
graphicInfo.WindowLocation = aaGraphic.WindowLocation.TopLeftCorner;
graphicInfo.WindowRelativePosition = aaGraphic.WindowRelativePosition.ClientAreaXY;
graphicInfo.RelativeTo = aaGraphic.RelativeTo.CustomizedWidthHeight;
graphicInfo.X = Boundary.X;
graphicInfo.Y = Boundary.Y;
graphicInfo.Height = Boundary.Height;
ShowGraphic( graphicInfo );
```





FIGURE 17: ACTION SCRIPT ON LEFT CLICK/KEY DOWN: SHOW GRAPHIC BUTTON

8. Select an Animation Action Script and type the script as shown below **On Left Click/Key Down**.

```
HideGraphic( "MyAnalogMeter" );
```





**FIGURE 18: ACTION SCRIPT ON LEFT CLICK/KEY DOWN: HIDE GRAPHIC BUTTON**

9. Create another InTouch Window called **Win4**.
10. Embed the above Symbol on the InTouch Window (Figure 19 below).




FIGURE 19: InTouch WINDOW

11. Switch to Runtime.
12. Click the **ShowGraphics** button. The analog meter is displayed. In the display the max value is **110**, min value is 10 and the actual data value is changing as written in the application script. This means that the original custom property values are overwritten using the **CustomPropertyValuePair** method, which is part of the **ShowGraphic** script function.




FIGURE 20: SHOW GRAPHIC WITH CUSTOMPROPERTYVALUEPAIR

13. Click the **HideGraphic** button. The Analog meter will disappear.

Lastly you can add another InTouch Window called **Main** and have 5 InTouch buttons where each button shows Win1, Win2, Win3 and Win4 using the ShowWindow animation. The last button is **CloseAll**, which hides all the windows using the HideWindow animation (Figure 21 below).



**FIGURE 21: MAIN INTOUCH WINDOW**

[Download and extract a published application](#) and run InTouch WindowMaker and InTouch WindowViewer. Note that you will not be able to see the Arcestra graphic scripting or the Arcestra Symbol in the IDE since this is a published application. You can copy/paste all the scripts included in this *Tech Note* and create a duplicate of the same Managed application as the application delivered in the downloaded zip file.

B. Shah

*Tech Notes* are published occasionally by Wonderware Technical Support. Publisher: Invensys Systems, Inc., 26561 Rancho Parkway South, Lake Forest, CA 92630. There is also technical information on our software products at [Wonderware Technical Support](#).

For technical support questions, send an e-mail to [wwsupport@invensys.com](mailto:wwsupport@invensys.com).

 [Back to top](#)

©2012 Invensys Systems, Inc. All rights reserved. No part of the material protected by this copyright may be reproduced or utilized in any form or by any means, electronic or mechanical, including photocopying, recording, broadcasting, or by any information storage and retrieval system, without permission in writing from Invensys Systems, Inc. [Terms of Use](#).